APPENDIX A.

```
10    !  Variables
20    !  Where possible a variable used in the following
30    !  routines is either the same as that used in the
40    !  formulae or can be deduced from the suffix.
50    !
60    !  No suffix = Upper Case Letter
70    !  Suffix 1 = Lower Case Letter
80    !  Suffix 2 = Lower Case Letter Squared
90    !  Suffix 0 = Subscript 0 e.g. E0 = Grid Eastings of True Origin.
100   !
110   !  J3—J9 are used for intermediate values.
120   !
130   !  The variables not covered by the above rules are listed below:
140   !  K = Phi (Latitude) or Phi'
150   !  L = Lambda (Longitude)
160   !  R = Rho (Radius of Curvature in Meridian)
170   !  V = Nu (Radius of Curvature in Prime Vertical)
180   !  H2 = Eta Squared (Nu/Rho - 1)
190   !  K3 = Phi2 - Phi1 (Difference Latitude)
200   !  K4 = Phi2 + Phi1 (Sum Latitudes)
210   !  Ga, Gb = (t - T) at line terminals A and B
220   !  All angular arguments are in Radians
230   !
240   !
250   !  Arc of Meridian
260   J3 = (1 + N1 + 5/4*N1^2 + 5/4*N1^3)*K3
270   J4 = (3*N1 + 3*N1^2 + 21/8*N1^3)*SIN(K3)*COS(K4)
280   J5 = (15/8*N1^2 + 15/8*N1^3)*SIN(2*K3)*COS(2*K4)
290   J6 = 35/24*N1^3*SIN(3*K3)*COS(3*K4)
300   M = B1*(J3 - J4 + J5 - J6)
310   RETURN
320   !
330   !
340   !  Compute Phi' (K)
350   K = (N - N0)/A1 + K0
360   K3 = K - K0
370   K4 = K + K0
380   GOSUB 260
390   IF ABS(N - N0 - M) < .001 THEN 420
400   K = K + (N - N0 - M)/A1
410   GOTO 360
420   RETURN
430   !
440   !
450   !  Compute V, R&H2
460   V = A1/SQR(1 - E2*SIN(K)^2)
470   R = V*(1 - E2)/(1 - E2*SIN(K)^2)
480   H2 = V/R - 1
490   RETURN
```

```
500  !
510  !
520  !  E&N from Latitude (K) & Longitude (L)
530  K3 = K - K0
540  K4 = K + K0
550  GOSUB 260
560  GOSUB 460
570  P = L - L0
580  J3 = M + N0                                                    !  I
590  J4 = V/2*SIN(K)*COS(K)                                         !  II
600  J5 = V/24*SIN(K)*COS(K)^3*(5 - TAN(K)^2 + 9*H2)                !  III
610  J6 = V/720*SIN(K)*COS(K)^5*(61 - 58*TAN(K)^2 + TAN(K)^4)       !  IIIA
620  N = J3 + P^2*J4 + P^4*J5 + P^6*J6
630  J7 = V*COS(K)                                                  !  IV
640  J8 = V/6*COS(K)^3*(V/R - TAN(K)^2)                             !  V
650  J9 = V/120*COS(K)^5
660  J9 = J9*(5 - 18*TAN(K)^2 + TAN(K)^4 + 14*H2 - 58*TAN(K)^2*H2)  !  VI
670  E = E0 + P*J7 + P^3*J8 + P^5*J9
680  RETURN
690  !
700  !
710  !  Latitude & Longitude from E & N
720  GOSUB 350
730  GOSUB 460
740  Y1 = E - E0
750  J3 = TAN(K)/(2*R*V)                                            !  VII
760  J4 = TAN(K)/(24*R*V^3)*(5 + 3*TAN(K)^2 + H2 - 9*TAN(K)^2*H2)   !  VIII
770  J5 = TAN(K)/(720*R*V^5)*(61 + 90*TAN(K)^2 + 45*TAN(K)^4)       !  IX
780  K9 = K - Y1^2*J3 + Y1^4*J4 - Y1^6*J5
790  J6 = 1/(COS(K)*V)                                              !  X
800  J7 = 1/(COS(K)*6*V^3)*(V/R + 2*TAN(K)^2)                       !  XI
810  J8 = 1/(COS(K)*120*V^5)*(5 + 28*TAN(K)^2 + 24*TAN(K)^4)        !  XII
820  J9 = 1/(COS(K)*5040*V^7)
830  J9 = J9*(61 + 662*TAN(K)^2 + 1320*TAN(K)^4 + 720*TAN(K)^6)     !  XIIA
840  L = L0 + Y1*J6 - Y1^3*J7 + Y1^5*J8 - Y1^7*J9
850  K = K9
860  RETURN
870  !
880  !
890  !  C from Latitude & Longitude
900  GOSUB 460
910  P = L - L0
920  J3 = SIN(K)                                                    !  XIII
930  J4 = SIN(K)*COS(K)^2/3*(1 + 3*H2 + 2*H2^2)                     !  XIV
940  J5 = SIN(K)*COS(K)^4/15*(2 - TAN(K)^2)                         !  XV
950  C = P*J3 + P^3*J4 + P^5*J5
960  RETURN
970  !
```

```
980  !
990  !  C from E & N
1000 GOSUB 350
1010 GOSUB 460
1020 Y1 = E - E0
1030 J3 = TAN(K)/V                                                    ! XVI
1040 J4 = TAN(K)/(3*V^3)*(1 + TAN(K)^2 - H2 - 2*H2^2)                 ! XVII
1050 J5 = TAN(K)/(15*V^5)*(2 + 5*TAN(K)^2 + 3*TAN(K)^4)               ! XVIII
1060 C = Y1*J3 - Y1^3*J4 + Y1^5*J5
1070 RETURN
1080 !
1090 !
1100 !  F from Latitude & Longitude
1110 GOSUB 460
1120 P = L - L0
1130 J3 = COS(K)^2/2*(1 + H2)                                         ! XIX
1140 J4 = COS(K)^4/24*(5 - 4*TAN(K)^2 + 14*H2 - 28*TAN(K)^2*H2)       ! XX
1150 F = F0*(1 + P^2*J3 + P^4*J4)
1160 RETURN
1170 !
1180 !
1190 !  F from E & N
1200 GOSUB 350
1210 GOSUB 460
1220 Y1 = E - E0
1230 J3 = 1/(2*R*V)                                                   ! XXI
1240 J4 = (1 + 4*H2)/(24*R^2*V^2)                                     ! XXII
1250 F = F0*(1 + Y1^2*J3 + Y1^4*J4)
1260 RETURN
1270 !
1280 !
1290 !  (t - T) from E, N
1300 N = (Na + Nb)/2
1310 GOSUB 350
1320 GOSUB 460
1330 J3 = 1/(6*R*V)                                                   ! XXIII
1340 Ga = (2*Y1a + Y1b)*(Na - Nb)*J3
1350 Gb = (2*Y1b + Y1a)*(Nb - Na)*J3
1360 RETURN
```